

```
(%i70) kill(all);
```

```
(%o0) done
```

relative representation of b and h

h is kept dimensional

b is defined through the $\text{upsilon} = b/h$ ratio

t is defined through the $\text{epsilon} = t/h$ ratio

```
(%i1) b : h * upsilon;
```

```
(%o1) h v
```

```
(%i2) t : h * epsilon;
```

```
(%o2)  $\epsilon h$ 
```

Dimensions are declared positive, in order to simplify $\text{abs}(x)$ with x

```
(%i3) assume(upsilon>0,epsilon>0,h>0);
```

```
(%o3) [ $v > 0, \epsilon > 0, h > 0$ ]
```

for the sake of simplicity, wall thickness is assumed constant.

two coordinate systems are employed:

OXY is centered at the web midpoint

Gxy is centered at the (initially unknown) section center of gravity

profile nodal point coordinates according to the OXY reference system

```
(%i4) XA : b $ YA: h/2$
      XB : 0 $ YB: h/2$
      XC : 0 $ YC:-h/2$
      XD : b $ YD:-h/2$
```

profile segment lengths

```
(%i12) l_AB : sqrt((XA-XB)^2 + (YA-YB)^2)$
      l_BC : sqrt((XB-XC)^2 + (YB-YC)^2)$
      l_CD : sqrt((XC-XD)^2 + (YC-YD)^2)$
```

(x,y) coordinates for points along the profile segments are parametrically defined with respect to a normalized arlength coordinate xi that spans from 0 to 1 while moving along the segment from the first to the second endpoint. The Gxy centroidal coordinate system is here employed, with unknown XG,YG.

```
(%i15) x_AB(xi) := (XB-XG)*xi + (XA-XG)*(1-xi)$
      y_AB(xi) := (YB-YG)*xi + (YA-YG)*(1-xi)$
      x_BC(xi) := (XC-XG)*xi + (XB-XG)*(1-xi)$
      y_BC(xi) := (YC-YG)*xi + (YB-YG)*(1-xi)$
      x_CD(xi) := (XD-XG)*xi + (XC-XG)*(1-xi)$
      y_CD(xi) := (YD-YG)*xi + (YC-YG)*(1-xi)$
```

Area calculation by integration along the section

```
(%i21) area : integrate(t * l_AB ,xi,0,1)
          + integrate(t * l_BC ,xi,0,1)
          + integrate(t * l_CD ,xi,0,1);
```

(%o21) $2 \varepsilon h^2 v + \varepsilon h^2$

First order (or "statical") moments of area, with respect to the Center of Gravity

```
(%i22) Mx : integrate(y_AB(xi) * t * l_AB ,xi,0,1)
          + integrate(y_BC(xi) * t * l_BC ,xi,0,1)
          + integrate(y_CD(xi) * t * l_CD ,xi,0,1);
```

(%o22)
$$-\frac{\varepsilon h^2 (h+2 YG) v}{2} + \frac{\varepsilon h^2 (h-2 YG) v}{2} - YG \varepsilon h^2$$

```
(%i23) My : integrate(x_AB(xi) * t * l_AB ,xi,0,1)
          + integrate(x_BC(xi) * t * l_BC ,xi,0,1)
          + integrate(x_CD(xi) * t * l_CD ,xi,0,1);
```

(%o23) $\varepsilon h^2 v (h v - 2 XG) - XG \varepsilon h^2$

the first order moment of area is known to be null
if the reference system is centroidal; such condition is employed
to evaluate the center of gravity coordinates with respect to OXY

```
(%i24) linsolve([Mx=0,My=0],[XG,YG]), globalsolve=true;
```

```
(%o24) [XG:  $\frac{h u^2}{2 u + 1}$ , YG: 0]
```

second order moments of area, with respect to Gxy

```
(%i25) Jxx : fullratsimp(
      integrate(y_AB(xi)^2 * t * l_AB ,xi,0,1)
    + integrate(y_BC(xi)^2 * t * l_BC ,xi,0,1)
    + integrate(y_CD(xi)^2 * t * l_CD ,xi,0,1)
  );
```

```
(%o25)  $\frac{6 \varepsilon h^4 u + \varepsilon h^4}{12}$ 
```

```
(%i26) Jyy : fullratsimp(
      integrate(x_AB(xi)^2 * t * l_AB ,xi,0,1)
    + integrate(x_BC(xi)^2 * t * l_BC ,xi,0,1)
    + integrate(x_CD(xi)^2 * t * l_CD ,xi,0,1)
  );
```

```
(%o26)  $\frac{\varepsilon h^4 u^4 + 2 \varepsilon h^4 u^3}{6 u + 3}$ 
```

```
(%i27) Jxy : integrate(x_AB(xi)*y_AB(xi) * t * l_AB ,xi,0,1)
    + integrate(x_BC(xi)*y_BC(xi) * t * l_BC ,xi,0,1)
    + integrate(x_CD(xi)*y_CD(xi) * t * l_CD ,xi,0,1);
```

(%o27) 0

due to the symmetric nature of the cross section, the product (not "mixed") moment of area is zero.

we now define the alpha,beta coefficients of the equation

$$\frac{d \sigma_z}{dz} = \alpha(x, y, \dots) S_y - \beta(x, y, \dots) S_x$$

Material is assumed homogeneous and isotropic along the section.

(%i28) define(alpha(x,y) , fullratsimp ((-Jxy * x + Jyy*y)/(Jxx*Jyy-Jxy^2)));

(%o28) $\alpha(x, y) := \frac{12 y}{6 \varepsilon h^4 u + \varepsilon h^4}$

(%i29) define(beta(x,y) , fullratsimp ((-Jxx * x + Jxy*y)/(Jxx*Jyy-Jxy^2)));

(%o29) $\beta(x, y) := -\frac{(6 u + 3) x}{\varepsilon h^4 u^4 + 2 \varepsilon h^4 u^3}$

for each point along the section, the axial stress component variation with increasing z is defined.
Such variation causes a disequilibrium that induces the compensating shear actions.

```
(%i30) define( dsigmaz_dz(x,y) , fullratsimp( alpha(x,y) * Sy - beta(x,y) * Sx ) );
```

```
(%o30) dsigmaz_dz(x,y) := 
$$\frac{(12 S_y u^4 + 24 S_y u^3) y + (36 S_x u^2 + 24 S_x u + 3 S_x) x}{6 \varepsilon h^4 u^5 + 13 \varepsilon h^4 u^4 + 2 \varepsilon h^4 u^3}$$

```

shear stress at the A end, which is a free surface in the case of an open section. In the case of a closed thin wall section, such value is generally nonzero, and it is to be treated as an unknown parameter.

```
(%i31) tau_A : 0;
```

```
(%o31) 0
```

AB segment : The tau_iz shear stress component is evaluated as a function of the normalized arc length coordinate xi, based on the axial equilibrium of the portion of section spanning from A (s=0) to the current point s=xi*l_AB along the segment

```
(%i32) define(
    tau_AB(xi),
    fullratsimp(
        (
            tau_A * t
            + integrate(
                dsigmaz_dz(x_AB(dummy), y_AB(dummy)) * t * l_AB,
                dummy, 0, xi
            )
        )/t
    )
);
```

$$(\%o32) \text{tau_AB}(\xi) := -\frac{\left(36 S_x v^2 + 24 S_x v + 3 S_x\right) \xi^2 + \left(-12 S_y v^3 + (-24 S_y - 36 S_x) v^2 - 42 S_x v - 6 S_x\right) \xi}{12 \varepsilon h^2 v^3 + 26 \varepsilon h^2 v^2 + 4 \varepsilon h^2 v}$$

BC segment: the shear stress component is evaluated as for segment AB;
at the B segment end, the shear stress transmitted from the previous segment is applied.

The same normalized arc length coordinate is considered, whereas the absolute arclength coordinate (if needed) may be defined as

$$s = l_{AB} + \xi l_{BC}, \quad ds = l_{BC} \cdot d\xi$$

```
(%i33) define(
    tau_BC(xi),
    fullratsimp(
        (
            tau_AB(1) * t
            + integrate(
                dsigmaz_dz(x_BC(dummy), y_BC(dummy)) * t * l_BC,
                dummy, 0, xi
            )
        )/t
    )
);
```

$$(\%o33) \text{tau_BC}(\xi) := -\frac{\left(12 S_y v^2 + 24 S_y v\right) \xi^2 + \left(-12 S_y v^2 + (36 S_x - 24 S_y) v + 6 S_x\right) \xi - 12 S_y v^3 - 24 S_y v^2 - 18 S_x v - 3 S_x}{12 \varepsilon h^2 v^3 + 26 \varepsilon h^2 v^2 + 4 \varepsilon h^2 v}$$

BC segment: the shear stress component is evaluated as for segments AB and CD.

here,

$$s = l_{AB} + l_{BC} + \xi \cdot l_{CD}, \quad ds = l_{CD} \cdot d\xi$$

```
(%i34) define(
      tau_CD(xi),
      fullratsimp(
        (
          tau_BC(1) * t
          + integrate(
            dsigmaz_dz(x_CD(dummy), y_CD(dummy)) * t * l_CD,
            dummy, 0, xi
          )
        )
      )/t
    );
```

```
(%o34) tau_CD(ξ):=
```

$$\frac{(36 S_x u^2 + 24 S_x u + 3 S_x) \xi^2 + (-12 S_y u^3 + (-24 S_y - 36 S_x) u^2 - 6 S_x u) \xi + 12 S_y u^3 + 24 S_y u^2 - 18 S_x u - 3 S_x}{12 \varepsilon h^2 u^3 + 26 \varepsilon h^2 u^2 + 4 \varepsilon h^2 u}$$

check : is tau_iz equal to zero at the D end??

```
(%i35) tau_D : fullratsimp(tau_CD(1));
```

```
(%o35) 0
```


wow. ok.

further check : let's plot the tau_zs variation along the cross section segments.
sample dimensions are required for extracting numerical results to be plotted.


```
(%i36) dim : [h=80,upsilon=40/80,epsilon=6/80];
```

```
(%o36) [h=80,upsilon= $\frac{1}{2}$ ,epsilon= $\frac{3}{40}$ ]
```

Shear resultants S_x, S_y are applied for plotting purposes s.t. the nominal shear stress, evaluated as the load divided by the area, is unitary.


```
(%i37) wxplot2d(
    ev(
        [
            tau_AB(xi),
            tau_BC(xi),
            tau_CD(xi)
        ],dim, Sy=area*1, Sx=0, infeas
    ),
    [xi,0,1],
    [legend,"AB","BC","CD"]
);
```

(%t37)



(%o37)

```
(%i38) wxplot2d(  
    ev(  
        [tau_AB(xi),tau_BC(xi),tau_CD(xi)],  
        dim, Sy=0, Sx=area*1 ,infeval  
    ),  
    [xi,0,1],  
    [legend,"AB","BC","CD"]  
);
```



I evaluate the peak value for the actual shear stress vs. nominal stress ratio along the AB segment, and along the section.

```
(%i39) tmp : ev(
    tau_AB(xi),
    dim, Sy=0, Sx=area*1 ,infeval
) $

linsolve(
    diff(
        tmp,
        xi
    )=0,
```

```

        xi
    )$

    ev(tmp , % , numer);

```

(%o41) 2.7

in order to evaluate resultants, a unit vector is to be defined for each segment that orients the tau_iz actions along the plane.

```

vx = - diff ( x, s) = -diff (x, xi) / l;
vy = - diff ( y, s) = -diff (y, xi) / l;

```

those vectors are a function of xi (and should be defined accordingly) in the case of curved segments.

the negative sign is due to the opposite "inward" and "growing s" directions; tau_zi are oriented inward, whereas the

```

[ diff ( x, s) , diff ( y, s) ]

```

vector is oriented towards growing s

```

(%i42) vx_AB : - diff(x_AB(xi),xi)/l_AB;
      vy_AB : - diff(y_AB(xi),xi)/l_AB;

      vx_BC : - diff(x_BC(xi),xi)/l_BC;
      vy_BC : - diff(y_BC(xi),xi)/l_BC;

      vx_CD : - diff(x_CD(xi),xi)/l_CD;
      vy_CD : - diff(y_CD(xi),xi)/l_CD;

```

```
(%o42) 1
(%o43) 0
(%o44) 0
(%o45) 1
(%o46) -1
(%o47) 0
```

x component of the resultant force of the τ_{iz} stress distribution.

```
(%i48) Fx : integrate(
      vx_AB * tau_AB(xi) * t * l_AB
    + vx_BC * tau_BC(xi) * t * l_BC
    + vx_CD * tau_CD(xi) * t * l_CD,
      xi, 0, 1
    );
```

```
(%o48) Sx
```

y component of the resultant force of the τ_{iz} stress distribution.

```
(%i49) Fy : integrate(
      vy_AB * tau_AB(xi) * t * l_AB
    + vy_BC * tau_BC(xi) * t * l_BC
    + vy_CD * tau_CD(xi) * t * l_CD,
      xi, 0, 1
    )$
Fy : fullratsimp(Fy);
```

(%o50) S_y

axial z component of the resultant moment of the tau_iz stress distribution.

```
(%i51) MGz : integrate(
      (vx_AB * (-y_AB(xi)) + vy_AB * (+x_AB(xi))) * tau_AB(xi) * t * l_AB
    + (vx_BC * (-y_BC(xi)) + vy_BC * (+x_BC(xi))) * tau_BC(xi) * t * l_BC
    + (vx_CD * (-y_CD(xi)) + vy_CD * (+x_CD(xi))) * tau_CD(xi) * t * l_CD,
      xi, 0, 1
    )$
MGz : fullratsimp(MGz);
```

(%o52)
$$-\frac{12 S_y h u^3 + 4 S_y h u^2}{12 u^2 + 8 u + 1}$$

The evaluation follows of the shear center coordinates (e,f), based on an equal resultant moment condition with respect to the CoG.

```
(%i53) eq : MGz = Sx * (-f) + Sy * (+e);
```

(%o53)
$$-\frac{12 S_y h u^3 + 4 S_y h u^2}{12 u^2 + 8 u + 1} = S_y e - S_x f$$

```
(%i54) eq_res : rhs(eq) - lhs(eq);
```

$$(\%o54) \frac{12 S_y h u^3 + 4 S_y h u^2}{12 u^2 + 8 u + 1} - S_x f + S_y e$$

such equation holds for null shear components,

we now impose that such null residual condition is also constant for arbitrary varying S_x , S_y

```
(%i55) ev( eq_res, Sx=0, Sy=0);
```

```
(%o55) 0
```

```
(%i56) linsolve(
      [
        diff(eq_res,Sx)=0,
        diff(eq_res,Sy)=0
      ], [e,f] ), globalsolve=true;
```

$$(\%o56) [e: -\frac{12 h u^3 + 4 h u^2}{12 u^2 + 8 u + 1}, f: 0]$$

check based on comparison with known results from literature, e.g.

<https://theconstructor.org/structural-engg/analysis/shear-centre-with-examples/3677/>

```
(%i57) fullratsimp(
      ( -(XG + b^2*h^2*t/4/Jxx) )
      - ( e )
```

```
);
```

```
(%o57) 0
```

ok

the shear contribution to the elastic strain energy per unit beam length is evaluated in the following

```
(%i58) dU_dz :
      integrate(
          tau_AB(xi)^2/2/G * t* l_AB
        + tau_BC(xi)^2/2/G * t* l_BC
        + tau_CD(xi)^2/2/G * t* l_CD,
          xi,0,1
      );
```

```
(%o58) (480 sy^2 u^7+2640 sy^2 u^6+(5040 sy^2+432 sx^2) u^5+(3864 sy^2+1656 sx^2) u^4+
(1056 sy^2+2244 sx^2) u^3+(96 sy^2+1158 sx^2) u^2+228 sx^2 u+15 sx^2) / (1440 G ε h^2 u^6+6240 G ε
h^2 u^5+7720 G ε h^2 u^4+2080 G ε h^2 u^3+160 G ε h^2 u^2)
```

a nominal counterpart is defined,
that contains the three corrective chi factors

```
(%i59) dU_dz_nom : (chi_x * Sx^2 + chi_y * Sy^2 + chi_xy * Sx*Sy)/2/G/area;
```


$$(\%o59) \frac{Sy^2 \chi_{i_y} + Sx Sy \chi_{i_{xy}} + Sx^2 \chi_{i_x}}{2 G (2 \varepsilon h^2 u + \varepsilon h^2)}$$

```
(%i60) eq : dU_dz = dU_dz_nom;
```

$$(\%o60) (480 Sy^2 u^7 + 2640 Sy^2 u^6 + (5040 Sy^2 + 432 Sx^2) u^5 + (3864 Sy^2 + 1656 Sx^2) u^4 + (1056 Sy^2 + 2244 Sx^2) u^3 + (96 Sy^2 + 1158 Sx^2) u^2 + 228 Sx^2 u + 15 Sx^2) / (1440 G \varepsilon h^2 u^6 + 6240 G \varepsilon h^2 u^5 + 7720 G \varepsilon h^2 u^4 + 2080 G \varepsilon h^2 u^3 + 160 G \varepsilon h^2 u^2) = \frac{Sy^2 \chi_{i_y} + Sx Sy \chi_{i_{xy}} + Sx^2 \chi_{i_x}}{2 G (2 \varepsilon h^2 u + \varepsilon h^2)}$$

```
(%i61) eq_res : lhs(eq) - rhs(eq);
```

$$(\%o61) (480 Sy^2 u^7 + 2640 Sy^2 u^6 + (5040 Sy^2 + 432 Sx^2) u^5 + (3864 Sy^2 + 1656 Sx^2) u^4 + (1056 Sy^2 + 2244 Sx^2) u^3 + (96 Sy^2 + 1158 Sx^2) u^2 + 228 Sx^2 u + 15 Sx^2) / (1440 G \varepsilon h^2 u^6 + 6240 G \varepsilon h^2 u^5 + 7720 G \varepsilon h^2 u^4 + 2080 G \varepsilon h^2 u^3 + 160 G \varepsilon h^2 u^2) - \frac{Sy^2 \chi_{i_y} + Sx Sy \chi_{i_{xy}} + Sx^2 \chi_{i_x}}{2 G (2 \varepsilon h^2 u + \varepsilon h^2)}$$

since it is not allowed to differentiate any expression with respect to non-atomic terms as $Sx^2, Sy^2, Sx*Sy$, a workaround is applied in the following consisting in a change of variable.

```
(%i62) eq_res : ratsubst( Sxq , Sx^2 , eq_res )$
eq_res : ratsubst( Syq , Sy^2 , eq_res )$
eq_res : ratsubst( SxSy , Sx*Sy , eq_res )$
```

```
(%i65) eq_res;
```

```
(%o65) (960 Syq v8+5760 Syq v7+
(-720 Syq chi_y-720 SxSy chi_xy-720 Sxq chi_x+12720 Syq+864 Sxq) v6+
(-3120 Syq chi_y-3120 SxSy chi_xy-3120 Sxq chi_x+12768 Syq+3744 Sxq) v5+
(-3860 Syq chi_y-3860 SxSy chi_xy-3860 Sxq chi_x+5976 Syq+6144 Sxq) v4+
(-1040 Syq chi_y-1040 SxSy chi_xy-1040 Sxq chi_x+1248 Syq+4560 Sxq) v3+
(-80 Syq chi_y-80 SxSy chi_xy-80 Sxq chi_x+96 Syq+1614 Sxq) v2+258 Sxq v+15 Sxq) / (
v2)
```

check: no residual Sx, Sy terms within eq_res

```
(%i66) diff( eq_res , Sx);
      diff( eq_res , Sy);
```

```
(%o66) 0
```

```
(%o67) 0
```

ok

final value for the three corrective shear strain energy coefficients

```
(%i68) linsolve(
      [
        diff(eq_res, Sxq) =0,
        diff(eq_res, Syq) =0,
```

```
diff(eq_res, SxSy)=0
],
[ chi_x, chi_y, chi_xy]
), globalsolve=true;
```

```
(%o68) [ chi_x:  $\frac{24 u^4 + 96 u^3 + 138 u^2 + 78 u + 15}{20 u^4 + 80 u^3 + 80 u^2}$ , chi_y:  $\frac{240 u^4 + 480 u^3 + 300 u^2 + 72 u + 6}{180 u^2 + 60 u + 5}$ , chi_xy: 0
```

```
]
```

```
(%i69) ev([ chi_x, chi_y, chi_xy] , dim ,numer);
```

```
(%o69) [3.264, 2.4, 0]
```

Created with [wxMaxima](#).